CSCI-GA.3033-004
Graphics Processing Units (GPUs): Architecture and Programming

Lecture : Parallel Patterns

Most slides of this lecture are from:
• David Kirk/NVIDIA and
• Wen-mei W. Hwu
EUniversity of Illinois

Mohamed Zahran (aka Z)
mzahran@cs.nyu.edu
http://www.mzahran.com
Convolution
Convolution

• An Array operation
• Output data element = weighted sum of a collection of neighboring input elements.
• The weights are defined by an input mask array.
• Usually used as filters to transform signals (or pixels or ...) into more desirable form.
Convolution

N

\[
\begin{array}{cccccc}
1 & 2 & 3 & 4 & 5 & 6 & 7 \\
\end{array}
\]

P

\[
\begin{array}{cccccc}
 & & & & & 57 & \\
\end{array}
\]

Mask

\[
\begin{array}{cccccc}
3 & 4 & 5 & 4 & 3 \\
\end{array}
\]

\[
\begin{array}{cccccc}
 & & & & & \\
3 & 8 & 15 & 16 & 15 \\
\end{array}
\]
Convolution

Convolution can also be 2D.
Convolution

N

P

M

1 2 3 4 5 6 7
2 3 4 5 6 7 8
3 4 5 6 7 8 9
4 5 6 7 8 5 6
5 6 7 8 5 6 7
6 7 8 9 0 1 2
7 8 9 0 1 2 3

321

1 2 3 2 1
2 3 4 3 2
3 4 5 4 3
2 3 4 3 2
1 2 3 2 1

1 4 9 8 5
4 9 16 15 12
9 16 25 24 21
8 15 24 21 16
5 12 21 16 5
Convolution

• Thread organized as 1D grid.
• Pvalue allows intermediate values to be accumulated in registers to save DRAM bw.
• We assume ghost values are 0.
• There will be control flow divergence (due to ghost elements).
• Ratio of floating point arithmetic calculation to global memory access is \( \sim 1.0 \) → What can we do??

```c
__global__ void convolution_1D_basic_kernel(float *N, float *M, float *P,
                                        int Mask_Width, int Width) {

    int i = blockIdx.x*blockDim.x + threadIdx.x;

    float Pvalue = 0;
    int N_start_point = i - (Mask_Width/2);
    for (int j = 0; j < Mask_Width; j++) {
        if (N_start_point + j >= 0 && N_start_point + j < Width) {
            Pvalue += N[N_start_point + j]*M[j];
        }
    }
    P[i] = Pvalue;
}
```
Regarding Mask M

- Size of M is typically small.
- The contents of M do not change during execution.
- All threads need to access M and in the same order.

Doesn’t this make M a good candidate for constant memory?
Constant Memory

• Constant memory variables are visible to all thread blocks.
• Constant memory variables cannot be changed during kernel execution.
• The size of constant memory can vary from device to device.
How to Use Constant Memory

• Host code allocates, initializes variables the same way as any other variables that need to be copied to the device

• Use `cudaMemcpyToSymbol(dest, src, size)` to copy the variable into the device memory

• This copy function tells the device that the variable will not be modified by the kernel and can be safely cached.
Mask M and Constant Memory

• In host:
  • `#define MASK_WIDTH 10`
    `__constant__ float M[MASK_WIDTH]`
  • Allocate and initialize a mask `h_M`
    • `cudaMemcpyToSymbol(M, h_M, MASK_WIDTH * sizeof(float), offset, kind);`

• Kernel functions
  – access constant memory variables as global variables → no need to pass pointers of these variables to the kernel as parameter.
Question: Isn’t the constant memory also in DRAM? Why is it assumed faster than global memory?

Answer:

• CUDA runtime knows that constant memory variables are not modified.
• It directs the hardware to aggressively cache them during kernel execution.
Tiled Convolution
Tiled 1D Convolution Basic Idea

Tile 0
- Ghost: 0 1 2 3
- Remaining: 4 5

Tile 1
- Remaining: 2 4 5 6 7

Tile 2
- Remaining: 6 8 9 10 11

Tile 3
- Remaining: 10 11 12 13 14 15

Halo points to:
- Tile 0
- Tile 1
- Tile 2
- Tile 3
int n = Mask_Width/2;
int halo_index_left = (blockIdx.x - 1)*blockDim.x + threadIdx.x;
if (threadIdx.x >= blockDim.x - n) {
    N_ds[threadIdx.x - (blockDim.x - n)] =
        (halo_index_left < 0) ? 0 : N[halo_index_left];
}
Loading the internal elements

\[
N_{ds}[n + threadIdx.x] = N[blockIdx.x*blockDim.x + threadIdx.x];
\]
int halo_index_right = (blockIdx.x + 1) * blockDim.x + threadIdx.x;
if (threadIdx.x < n) {
    N_ds[n + blockDim.x + threadIdx.x] =
        (halo_index_right >= Width) ? 0 : N[halo_index_right];
}
__global__ void convolution_1D_tiled_kernel(float *N, float *P, int Mask_Width, int Width) {

    int i = blockIdx.x*blockDim.x + threadIdx.x;
    __shared__ float N_ds[TILE_SIZE + MAX_MASK_WIDTH - 1];

    int n = Mask_Width/2;

    int halo_index_left = (blockIdx.x - 1)*blockDim.x + threadIdx.x;
    if (threadIdx.x >= blockDim.x - n) {
        N_ds[threadIdx.x - (blockDim.x - n)] =
        (halo_index_left < 0) ? 0 : N[halo_index_left];
    }

    N_ds[n + threadIdx.x] = N[blockIdx.x*blockDim.x + threadIdx.x];

    int halo_index_right = (blockIdx.x + 1)*blockDim.x + threadIdx.x;
    if (threadIdx.x < n) {
        N_ds[n + blockDim.x + threadIdx.x] =
        (halo_index_right >= Width) ? 0 : N[halo_index_right];
    }

    __syncthreads();

    float Pvalue = 0;
    for(int j = 0; j < Mask_Width; j++) {
        Pvalue += N_ds[threadIdx.x + j]*M[j];
    }
    P[i] = Pvalue;
}
Shared Memory Data Reuse

- Element 2 is used by thread 4 (1X)
- Element 3 is used by threads 4, 5 (2X)
- Element 4 is used by threads 4, 5, 6 (3X)
- Element 5 is used by threads 4, 5, 6, 7 (4X)
- Element 6 is used by threads 4, 5, 6, 7 (4X)
- Element 7 is used by threads 5, 6, 7 (3X)
- Element 8 is used by threads 6, 7 (2X)
- Element 9 is used by thread 7 (1X)

Mask_Width is 5
Ghost Cells

N


P[0]
P[1]
P[2]
P[3]
P[4]
P[5]
P[6]
__global__ void convolution_1D_tiled_cache_kernel(float *N, float *P,
int Mask_Width, int Width) {

int i = blockIdx.x*blockDim.x + threadIdx.x;
__shared__ float N_ds[TILE_SIZE];
N_ds[threadIdx.x] = N[i];
__syncthreads();

int This_tile_start_point = blockIdx.x * blockDim.x;
int Next_tile_start_point = (blockIdx.x + 1) * blockDim.x;
int N_start_point = i - (Mask_Width/2);
float Pvalue = 0;
for (int j = 0; j < Mask_Width; j ++) {
    int N_index = N_start_point + j;
    if (N_index >= 0 && N_index < Width) {
        if ((N_index >= This_tile_start_point)
            && (N_index < Next_tile_start_point)) {
            Pvalue += N_ds[threadIdx.x+j-(Mask_Width/2)]*M[j];
        } else {
            Pvalue += N[N_index] * M[j];
        }
    }
}
P[i] = Pvalue;
}
Tiling $P$

- Use a thread block to calculate a tile of $P$
  - Thread Block size determined by the TILE_SIZE
Tiling N

- Each N element is used in calculating up to KERNEL_SIZE * KERNEL_SIZE elements (all elements in the tile)
High-Level Tiling Strategy

• Load a tile of $N$ into shared memory (SM)
  - All threads participate in loading
  - A subset of threads then use each $N$ element in SM
Output Tiling and Thread Index (P)

- Use a thread block to calculate a tile of P
  - Each output tile is of TILE_SIZE for both x and y

  \[
  \text{col}_o = \text{blockIdx}.x \times \text{TILE}_\text{SIZE} + tx; \\
  \text{row}_o = \text{blockIdx}.y \times \text{TILE}_\text{SIZE} + ty;
  \]
Tiling N

- Each N element is used in calculating up to KERNEL_SIZE * KERNEL_SIZE P elements (all elements in the tile)
Input tiles need to be larger than output tiles.

We will use a strategy where the input tile will be loaded into the shared memory.
Dealing with Mismatch

- Use a thread block that matches input tile
  - Each thread loads one element of the input tile
  - Some threads do not participate in calculating output
    - There will be if statements and control divergence
Shifting from output coordinates to input coordinates
Shifting from output coordinates to input coordinate

```c
int tx = threadIdx.x;
int ty = threadIdx.y;
int row_o = blockIdx.y * TILE_SIZE + ty;
int col_o = blockIdx.x * TILE_SIZE + tx;

int row_i = row_o - 2;
int col_i = col_o - 2;
```
Threads that loads halos outside N should return 0.0
Taking Care of Boundaries

```cpp
float output = 0.0f;

if((row_i >= 0) && (row_i < N.height) &&
   (col_i >= 0) && (col_i < N.width) ) {
   Ns[ty][tx] = N.elements[row_i*N.width + col_i];
} else{
   Ns[ty][tx] = 0.0f;
}
```
Some threads do not participate in calculating output.

```c
if(ty < TILE_SIZE && tx < TILE_SIZE){
    for(i = 0; i < 5; i++) {
        for(j = 0; j < 5; j++) {
            output += Mc[i][j] * Ns[i+ty][j+tx];
        }
    }
}
```
Some threads do not write output

```c
if(row_o < P.height && col_o < P.width)
    P.elements[row_o * P.width + col_o] = output;
}
```
Setting Block Size

#define BLOCK_SIZE (TILE_SIZE + 4)

dim3
dimBlock(BLOCK_SIZE,BLOCK_SIZE);

In general, block size should be

tile size + (kernel size -1)

Dim3 dimGrid(N.width/TILE_SIZE,
N.height/TILE_SIZE, 1)
In General

• **BLOCK_SIZE** is limited by the maximal number of threads in a thread block

• Input tile sizes could be

  \[ N \times \text{TILE\_SIZE} + (\text{KERNEL\_SIZE}-1) \]

  – By having each thread to calculate \(N\) input points (thread coarsening)
  – \(N\) is limited by the shared memory size

• **KERNEL\_SIZE** is decided by application needs
Reduction Trees
What? And Why?

• Arguably the most widely used parallel computation pattern.
• A commonly used strategy for processing large input data sets
  – There is no required order of processing elements in a data set (associative and commutative)
  – Partition the data set into smaller chunks
  – Have each thread to process a chunk
  – Use a reduction tree to summarize the results from each chunk into the final answer
• We will focus on the reduction tree step for now.
• Google and Hadoop MapReduce frameworks are examples of this pattern
Reduction enables other techniques

• Reduction is also needed to clean up after some commonly used parallelizing transformations

• Example: Privatization
  – Multiple threads write into an output location
  – Replicate the output location so that each thread has a private output location
  – Use a reduction tree to combine the values of private locations into the original output location
What is a reduction computation

• Summarize a set of input values into one value using a “reduction operation”
  – Max
  – Min
  – Sum
  – Product
  – Often with user defined reduction operation function as long as the operation
    • Is associative and commutative
    • Has a well-defined identity value (e.g., 0 for sum)
An efficient sequential reduction algorithm performs $N$ operations in $O(N)$

- Initialize the result as an identity value for the reduction operation
  - Smallest possible value for max reduction
  - Largest possible value for min reduction
  - 0 for sum reduction
  - 1 for product reduction

- Scan through the input and perform the reduction operation between the result value and the current input value
A parallel reduction tree algorithm performs $N-1$ Operations in $\log(N)$ steps.
A tournament is a reduction tree with “max” operation.

A more artful rendition of the reduction tree.
A Quick Analysis

• For N input values, the reduction tree performs
  – \((\frac{1}{2})N + (\frac{1}{4})N + (\frac{1}{8})N + \ldots + (\frac{1}{N}) = (1 - (\frac{1}{N}))N = N - 1\) operations
  – In \(\log(N)\) steps - 1,000,000 input values take 20 steps
    • Assuming that we have enough execution resources
  – Average Parallelism \((N-1)/\log(N))\)
    • For \(N = 1,000,000\), average parallelism is 50,000
    • However, peak resource requirement is 500,000!

• This is a work-efficient parallel algorithm
  – The amount of work done is comparable to sequential
  – Many parallel algorithms are not work efficient
A Sum Reduction Example

- **Parallel implementation:**
  - Recursively halve the # of threads, add two values per thread in each step
  - Takes log(n) steps for n elements, requires n/2 threads

- **Assume an in-place reduction using shared memory**
  - The original vector is in device global memory
  - The shared memory is used to hold a partial sum vector
  - Each step brings the partial sum vector closer to the sum
  - The final sum will be in element 0
  - Reduces global memory traffic due to partial sum values
## Vector Reduction with Branch Divergence

<table>
<thead>
<tr>
<th>Thread 0</th>
<th>Thread 1</th>
<th>Thread 2</th>
<th>Thread 3</th>
<th>Thread 4</th>
<th>Thread 5</th>
</tr>
</thead>
<tbody>
<tr>
<td>0</td>
<td>1</td>
<td>2</td>
<td>3</td>
<td>4</td>
<td>5</td>
</tr>
<tr>
<td>0+1</td>
<td>2+3</td>
<td>4+5</td>
<td>6+7</td>
<td>6+7</td>
<td>8+9</td>
</tr>
<tr>
<td>0...3</td>
<td>4..7</td>
<td>8..11</td>
<td>8..11</td>
<td>8..15</td>
<td>8..15</td>
</tr>
<tr>
<td>0..7</td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
</tbody>
</table>

*Partial Sum elements*
A Sum Example

Thread 0 | Thread 1 | Thread 2 | Thread 3
---|---|---|---
3 | 1 | 7 | 0
4 | 7 | 5 | 9
11 | 14 | | |
25 | | | |

Active Partial Sum elements
Simple Thread Index to Data Mapping

• Each thread is responsible of an even-index location of the partial sum vector
  – One input value is at the location of responsibility

• After each step, half of the threads are no longer needed

• In each step, one of the inputs comes from an increasing distance away
Optimizing Reduction Trees

• Performance factors of a reduction kernel
  – Memory coalescing
  – Control divergence
  – Thread utilization
A Sum Example (review)

<table>
<thead>
<tr>
<th>Thread 0</th>
<th>Thread 1</th>
<th>Thread 2</th>
<th>Thread 3</th>
</tr>
</thead>
<tbody>
<tr>
<td>3</td>
<td>1</td>
<td>7</td>
<td>0</td>
</tr>
<tr>
<td>7</td>
<td>0</td>
<td>4</td>
<td>1</td>
</tr>
<tr>
<td>0</td>
<td>4</td>
<td>1</td>
<td>6</td>
</tr>
<tr>
<td>4</td>
<td>1</td>
<td>6</td>
<td>3</td>
</tr>
</tbody>
</table>

Active Partial Sum elements

Steps:
1. 4
2. 7
3. 11
4. 25
The Reduction Steps

```c
for (unsigned int stride = 1;
    stride <= blockDim.x; stride *= 2)
{
    __syncthreads();
    if (t % stride == 0)
        partialSum[2*t] +=
        partialSum[2*t+stride];
}
```

Why do we need `syncthreads()`?
Barrier Synchronization

• `__syncthreads()` are needed to ensure that all elements of each version of partial sums have been generated before we proceed to the next step.

• Why do we not need another `__syncthread()` at the end of the reduction loop?
Back to the Global Picture

- At the end of the kernel execution, thread 0 in each block writes the sum of the block in `partialSum[0]` into a vector indexed by the value of `blockIdx.x`

- There can be a large number of such sums if the original vector is very large
  - The host code may iterate and launch another kernel

- If there are only a small number of sums, the host can simply transfer the data back and add them together.
Some Observations

• In each iteration, two control flow paths will be sequentially traversed for each warp
  – Threads that perform addition and threads that do not
  – Threads that do not perform addition still consume execution resources

• No more than half of threads will be executing after the first step
  – All odd-index threads are disabled after first step
  – After the 5th step, entire warps in each block will fail the if-condition, poor resource utilization but no divergence.
    • This can go on for a while, up to 5 more steps (1024/32=16= 2^5), where each active warp only has one productive thread until all warps in a block retire
  – Some warps will still succeed, but with divergence since only one thread will succeed
Thread Index Usage Matters

• In some algorithms, one can shift the index usage to improve the divergence behavior
  – Commutative and associative operators

• Reduction satisfies this criterion.
A Better Strategy

• Always compact the partial sums into the first locations in the partialSum[] array

• Keep the active threads consecutive
An Example of 16 threads

Thread 0  Thread 1  Thread 2  Thread 14  Thread 15

0  1  2  3  ...  13  14  15  16  17  18  19

0+16  15+31

...
A Better Reduction Kernel

for (unsigned int stride = blockDim.x;
    stride >= 1; stride /= 2)
{
    __syncthreads();
    if (t < stride)
        partialSum[t] +=
        partialSum[t+stride];
}
A Quick Analysis

• For a 1024 thread block
  – No divergence in the first 5 steps
  – 1024, 512, 256, 128, 64, 32 consecutive threads are active in each step
  – The final 5 steps will still have divergence
__shared__ float partialSum[2*BLOCK_SIZE];

unsigned int t = threadIdx.x;

unsigned int start = 2*blockIdx.x*blockDim.x;
partialSum[t] = input[start + t];
partialSum[blockDim+t] = input[start+ blockDim.x+t];

for (unsigned int stride = blockDim.x; stride >= 1; stride >>= 1)
{
    __syncthreads();
    if (t < stride)
        partialSum[t] += partialSum[t+stride];
}
Parallel Algorithm Overhead

__shared__ float partialSum[2*BLOCK_SIZE];

unsigned int t = threadIdx.x;
unsigned int start = 2*blockIdx.x*blockDim.x;
partialSum[t] = input[start + t];
partialSum[blockDim+t] = input[start+ blockDim.x+t];
for (unsigned int stride = blockDim.x;
    stride >= 1;  stride >>= 1)
{
    __syncthreads();
    if (t < stride)
    
        partialSum[t] += partialSum[t+stride];
}
Parallel Execution Overhead

```
3  1  7  0  4  1  6  3
```

```
+  +  +  +  +  +  +  +
```

```
4  7  5  9
```

```
+  +  +  +  +  +  +  +
```

```
7  6  7
```

```
+  +  +  +  +  +  +  +
```

```
7
```
Parallel Execution Overhead

Although the number of “operations” is N, each “operation involves much more complex address calculation and intermediate result manipulation.

If the parallel code is executed on a single-thread hardware, it would be significantly slower than the code based on the original sequential algorithm.
Parallel Scan (Prefix Sum)
What? Why?

• Frequently used for parallel work assignment and resource allocation
• A **key primitive** in many parallel algorithms to convert serial computation into parallel computation
  – Based on reduction tree and reverse reduction tree
(Inclusive) Scan (Prefix-Sum) Definition

**Definition:** The scan operation takes a binary associative operator $\oplus$, and an array of $n$ elements $[x_0, x_1, \ldots, x_{n-1}]$, and returns the prefix-sum array

$$[x_0, (x_0 \oplus x_1), \ldots, (x_0 \oplus x_1 \oplus \ldots \oplus x_{n-1})].$$

**Example:** If $\oplus$ is addition, then the scan operation on the array $[3 \ 1 \ 7 \ 0 \ 4 \ 1 \ 6 \ 3]$, would return $[3 \ 4 \ 11 \ 11 \ 15 \ 16 \ 22 \ 25]$. 
A Inclusive Scan Application Example

• Assume that we have a 100-inch bread to feed 10
• We know how much each person wants in inches
  – [3 5 2 7 28 4 3 0 8 1]
• How do we cut the bread quickly?
• How much will be left

• Method 1: cut the sections sequentially: 3 inches first, 5 inches second, 2 inches third, etc.

• Method 2: calculate prefix-sum array
  – [3, 8, 10, 17, 45, 49, 52, 52, 60, 61] (39 inches left)
Typical Applications of Scan

- **Scan is a simple and useful parallel building block**
  - Convert recurrences from sequential:
    \[
    \text{for}(j=1; j<n; j++) \quad \text{out}[j] = \text{out}[j-1] + f(j);
    \]
  - into parallel:
    \[
    \text{forall}(j) \quad \text{temp}[j] = f(j);
    \text{scan(out, temp)};
    \]

- **Useful for many parallel algorithms:**
  - radix sort
  - quicksort
  - String comparison
  - Lexical analysis
  - Stream compaction
  - Polynomial evaluation
  - Solving recurrences
  - Tree operations
  - Histograms
  - Etc.
Other Applications

- Assigning camp slots
- Assigning farmer market space
- Allocating memory to parallel threads
- Allocating memory buffer to communication channels
- ...
An Inclusive Sequential Scan

Given a sequence \([x_0, x_1, x_2, \ldots]\)

Calculate output \([y_0, y_1, y_2, \ldots]\)

Such that
\[
\begin{align*}
y_0 &= x_0 \\
y_1 &= x_0 + x_1 \\
y_2 &= x_0 + x_1 + x_2
\end{align*}
\]

\[\ldots\]

Using a recursive definition
\[
y_i = y_{i-1} + x_i
\]
An Sequential C Implementation

\[ y[0] = x[0]; \]
\[ \text{for} \ (i = 1; \ i < \text{Max}_i; \ i++) \ y[i] = y[i-1] + x[i]; \]

Computationally efficient: N additions needed for N elements - \( O(N) \)!
A Naïve Inclusive Parallel Scan

• Assign one thread to calculate each y element
• Have every thread to add up all x elements needed for the y element

\[
\begin{align*}
y_0 &= x_0 \\
y_1 &= x_0 + x_1 \\
y_2 &= x_0 + x_1 + x_2
\end{align*}
\]

“Parallel programming is easy as long as you do not care about performance.”
Parallel Inclusive Scan using Reduction Trees

- Calculate each output element as the reduction of all previous elements
  - Some reduction partial sums will be shared among the calculation of output elements
  - Based on design by Peter Kogge and Harold Stone at IBM in the 1970s - Kogge-Stone Trees
A Slightly Better Parallel Inclusive Scan Algorithm

1. Load input from global memory into shared memory array $T$

Each thread loads one value from the input (global memory) array into shared memory array $T$. 

| T | 3 | 1 | 7 | 0 | 4 | 1 | 6 | 3 |
1. (previous slide)

2. Iterate log(n) times, stride from 1 to ceil(n/2.0). Threads stride to n-1 active: add pairs of elements that are stride elements apart.

- Active threads: stride to n-1 (n-stride threads)
- Thread j adds elements j and j-stride from T and writes result into shared memory buffer T
- Each iteration requires two syncthreads
  - syncthreads(); // make sure that input is in place
  - float temp = T[j] + T[k - stride];
A Kogge-Stone Parallel Scan Algorithm

1. (previous slide)

2. Iterate \( \log(n) \) times, stride from 1 to \( \text{ceil}(n/2.0) \). Threads \textit{stride} to \( n-1 \) \textit{active}: add pairs of elements that are \textit{stride} elements apart.

- Active threads: \textit{stride} to \( n-1 \) (\( n \)-\textit{stride} threads)
- Thread \( j \) adds elements \( j \) and \( j-\text{stride} \) from \( T \) and writes result into shared memory buffer \( T \)
- Each iteration requires two syncthreads
  - \texttt{syncthreads();} // make sure that input is in place
  - float \( \text{temp} = T[j] + T[j - \text{stride}] \);
  - \texttt{syncthreads();} // make sure that previous output has been consumed
  - \( T[j] = \text{temp} \);

<table>
<thead>
<tr>
<th>3</th>
<th>1</th>
<th>7</th>
<th>0</th>
<th>4</th>
<th>1</th>
<th>6</th>
<th>3</th>
</tr>
</thead>
<tbody>
<tr>
<td>3</td>
<td>4</td>
<td>8</td>
<td>7</td>
<td>4</td>
<td>5</td>
<td>7</td>
<td>9</td>
</tr>
</tbody>
</table>

Iteration #1
Stride = 1
A Kogge-Stone Parallel Scan Algorithm

1. ...

2. Iterate log(n) times, stride from 1 to ceil(n/2.0). Threads stride to n-1 active: add pairs of elements that are stride elements apart.

Iteration #2
Stride = 2
1. Load input from global memory to shared memory.

2. Iterate log(n) times, stride from 1 to ceil(n/2.0). Threads *stride* to n-1 active: add pairs of elements that are *stride* elements apart.

3. Write output from shared memory to device memory
Double Buffering

- Use two copies of data T0 and T1
- Start by using T0 as input and T1 as output
- Switch input/output roles after each iteration
  - Iteration 0: T0 as input and T1 as output
  - Iteration 1: T1 as input and T0 and output
  - Iteration 2: T0 as input and T1 as output
- This is typically implemented with two pointers, source and destination that swap their contents from one iteration to the next
- This eliminates the need for the second syncthreads
1. (previous slide)

2. Iterate \( \log(n) \) times, stride from 1 to \( \lceil n/2.0 \rceil \). Threads \textit{stride} to \( n-1 \) active: add pairs of elements that are \textit{stride} elements apart.

- Active threads: \textit{stride} to \( n-1 \) (\( n\text{-stride} \) threads)
- Thread \( j \) adds elements \( j \) and \( j\text{-stride} \) from \( T \) and writes result into shared memory buffer \( T \)
- Each iteration requires only one \texttt{syncthreads}:
  - \texttt{syncthreads();} // make sure that input is in place
  - float destination[j] = source[j] + source[j - stride];
  - temp = destination; destination = source; source = temp;

Iteration #1
Stride = 1
Work Efficiency Analysis

• A Kogge-Stone scan kernel executes $\log(n)$ parallel iterations
  – The steps do $(n-1)$, $(n-2)$, $(n-4)$,..$(n- n/2)$ add operations each
  – Total # of add operations: $n * \log(n) - (n-1) \rightarrow O(n*\log(n))$ work

• This scan algorithm is not very work efficient
  – Sequential scan algorithm does $n$ adds
  – A factor of $\log(n)$ hurts: 20x for 1,000,000 elements!
  – Typically used within each block, where $n \leq 1,024$

• A parallel algorithm can be slow when execution resources are saturated due to low work efficiency
A Kogge-Stone Parallel Scan Algorithm

<table>
<thead>
<tr>
<th>T</th>
<th>3</th>
<th>1</th>
<th>7</th>
<th>0</th>
<th>4</th>
<th>1</th>
<th>6</th>
<th>3</th>
</tr>
</thead>
<tbody>
<tr>
<td>T</td>
<td>3</td>
<td>4</td>
<td>8</td>
<td>7</td>
<td>4</td>
<td>5</td>
<td>7</td>
<td>9</td>
</tr>
<tr>
<td>T</td>
<td>3</td>
<td>4</td>
<td>11</td>
<td>11</td>
<td>12</td>
<td>12</td>
<td>11</td>
<td>14</td>
</tr>
<tr>
<td>T</td>
<td>3</td>
<td>4</td>
<td>11</td>
<td>11</td>
<td>15</td>
<td>16</td>
<td>22</td>
<td>25</td>
</tr>
</tbody>
</table>

Stride = 1

Stride = 2

Stride = 4
Improving Efficiency

- A common parallel algorithm pattern: *Balanced Trees*
  - Build a balanced binary tree on the input data and sweep it to and from the root
  - Tree is not an actual data structure, but a concept to determine what each thread does at each step

- For scan:
  - Traverse down from leaves to root building partial sums at internal nodes in the tree
    - Root holds sum of all leaves
  - Traverse back up the tree building the scan from the partial sums
Brent-Kung Parallel Scan - Reduction Step

\[ \sum_{x_0}^{x_1} \quad \sum_{x_2}^{x_3} \quad \sum_{x_4}^{x_5} \quad \sum_{x_6}^{x_7} \]

In place calculation

Final value after reduce
Move (add) a critical value to a central location where it is needed.
Inclusive Post Scan Step

\[ x_0 \quad \sum_{x_0..x_1} \quad x_2 \quad \sum_{x_0..x_3} \quad x_4 \quad \sum_{x_4..x_5} \quad x_6 \quad \sum_{x_0..x_7} \]

\[ \sum_{x_0..x_2} \quad \sum_{x_0..x_4} \quad \sum_{x_0..x_6} \]
Reduction Step Kernel Code

// float T[BLOCK_SIZE] is in shared memory

int stride = 1;
while(stride < BLOCK_SIZE)
{
    int index = (threadIdx.x+1)*stride*2 - 1;
    if(index < BLOCK_SIZE)
        scan_array[index] += scan_array[index-stride];
    stride = stride*2;

    __syncthreads();
}
Reduction Step Kernel Code

// float T[BLOCK_SIZE] is in shared memory

int stride = 1;
while(stride < BLOCK_SIZE)
{
    int index = (threadIdx.x+1)*stride*2 - 1;
    if(index < BLOCK_SIZE)
        T[index] += T[index-stride];
    stride = stride*2;
    __syncthreads();
}

// threadIdx.x+1 = 1, 2, 3, 4...
// stride = 1, index =
Putting it together
Post Scan Step

```c
int stride = BLOCK_SIZE/2;
while(stride > 0)
{
    int index = (threadIdx.x+1)*stride*2 - 1;
    if(index < BLOCK_SIZE)
    {
        T[index+stride] += T[index];
    }
    stride = stride / 2;
    __syncthreads();
}
```
Work Analysis

• The parallel Inclusive Scan executes $2^* \log(n)$ parallel iterations
  – $\log(n)$ in reduction and $\log(n)$ in post scan
  – The iterations do $n/2$, $n/4$, ..., 1, 1, ..., n/4. n/2 adds
  – Total adds: $2^* (n-1) \rightarrow O(n)$ work

• The total number of adds is no more than twice of that done in the efficient sequential algorithm
  – The benefit of parallelism can easily overcome the 2X work when there is sufficient hardware
A couple of details

• Brent-Kung uses half the number of threads compared to Kogge-Stone
  – Each thread should load two elements into the shared memory

• Brent-Kung takes twice the number of steps compared to Kogge-Stone
  – Kogge-Stone is more popular for parallel scan with blocks in GPUs
Overall Flow of Complete Scan
A Hierarchical Approach

Initial Array of Arbitrary Values

Scan Block 0  |  Scan Block 1  |  Scan Block 2  |  Scan Block 3

Store Block Sum to Auxiliary Array

Scan Block Sums

Add Scanned Block Sum $i$ to All Values of Scanned Block $i + 1$

Final Array of Scanned Values
Using Global Memory Contents in CUDA

- Data in registers and shared memory of one thread block are not visible to other blocks.
- To make data visible, the data has to be written into global memory.
- However, any data written to the global memory are not visible until a memory fence. This is typically done by terminating the kernel execution.
- Launch another kernel to continue the execution. The global memory writes done by the terminated kernels are visible to all thread blocks.
Overall Flow of Complete Scan
A Hierarchical Approach

Initial Array of Arbitrary Values

- Scan Block 0
- Scan Block 1
- Scan Block 2
- Scan Block 3

Kernel

Store Block Sum to Auxiliary Array

Kernel

Scan Block Sums

Kernel

Add Scanned Block Sum $i$ to All Values of Scanned Block $i + 1$

Final Array of Scanned Values
Working on Arbitrary Length Input

• Build on the scan kernel that handles up to $2 \times \text{blockDim}.x$ elements
• For Kogge-Stone, have each section of blockDim.x elements assigned to a block
• Have each block write the sum of its section into a Sum array indexed by blockIdx.x
• Run parallel scan on the Sum array
  – May need to break down Sum into multiple sections if it is too big for a block
• Add the scanned Sum array values to the elements of corresponding sections
(Exclusive) Scan Definition

**Definition:** The exclusive scan operation takes a binary associative operator $\oplus$, and an array of $n$ elements $[x_0, x_1, \ldots, x_{n-1}]$ and returns the array $[0, x_0, (x_0 \oplus x_1), \ldots, (x_0 \oplus x_1 \oplus \ldots \oplus x_{n-2})]$.  

**Example:** If $\oplus$ is addition, then the exclusive scan operation on $[3 \ 1 \ 7 \ 0 \ 4 \ 1 \ 6 \ 3]$, would return $[0 \ 3 \ 4 \ 11 \ 11 \ 15 \ 16 \ 22]$.  


Why Exclusive Scan

• To find the beginning address of allocated buffers

• Inclusive and Exclusive scans can be easily derived from each other; it is a matter of convenience

Inclusive: [3 4 11 11 15 16 22]

Exclusive: [0 3 4 11 11 15 16 22 25]
An Exclusive Post Scan Step (Add-move Operation)
Exclusive Post Scan Step
Exclusive Post Scan Step

if (threadIdx.x == 0) T[2*blockDim.x-1] = 0;
int stride = BLOCK_SIZE;
while(stride > 0)
{
    int index = (threadIdx.x+1)*stride*2 - 1;
    if(index < 2* BLOCK_SIZE)
    {
        float temp = T[index+stride];
        T[index+stride] += T[index];
        T[index] = temp;
    }
    stride = stride / 2;
    __syncthreads();
}
Exclusive Scan Example - Reduction Step

Assume array is already in shared memory
Reduction Step (cont.)

Iteration 1, \( n/2 \) threads

Stride 1

Iterate \( \log(n) \) times. Each thread adds value \( \text{stride} \) elements away to its own value.

Each \( \oplus \) corresponds to a single thread.
Reduction Step (cont.)

Iterate log(n) times. Each thread adds value stride elements away to its own value.
Reduction Step (cont.)

Iterate $\log(n)$ times. Each thread adds value \textit{stride} elements away to its own value.

Note that this algorithm operates in-place: no need for double buffering.
Zero the Last Element

We now have an array of partial sums. Since this is an exclusive scan, set the last element to zero. It will propagate back to the first element.
Post Scan Step from Partial Sums

| T | 3 | 4 | 7 | 11 | 4 | 5 | 6 | 0 |
Iterate \( \log(n) \) times. Each thread adds value \textit{stride} elements away to its own value, and sets the value \textit{stride} elements away to its own \textit{previous} value.
Iterate $\log(n)$ times. Each thread adds value $stride$ elements away to its own value, and sets the value $stride$ elements away to its own previous value.
Done! We now have a completed scan that we can write out to device memory.

Total steps: $2 \times \log(n)$.
Total work: $2 \times (n-1)$ adds $= \mathcal{O}(n)$ Work Efficient!
A simpler exclusive scan kernel

- Adapt an inclusive, work in-efficient scan kernel
- Block 0:
  - Thread 0 loads 0 into XY[0]
  - Other threads load X[threadIdx.x-1] into XY[threadIdx.x]
- All other blocks:
  - All thread load X[blockIdx.x*blockDim.x+threadIdx.x-1] into XY[threadIdx.x]
- Similar adaption for Brent-Kung kernel but pay attention that each thread loads two elements
  - Only one zero should be loaded
  - All elements should be shifted by only one position
- Intellectual contribution vs. practical contribution
Conclusions

• We have reviewed several useful parallel patterns that you can use in your own GPU programming:
  – Convolution and tiled convolution
  – Reduction trees
  – Prefix scan (inclusive and exclusive)
• Parallel version must be work efficient
• Then we apply different GPU optimizations from our bag of tricks (coalescing, shared memory usage, ...).